
Abstract

This paper aims to show how (hardware) protocol verification can be
done using the higher-order theorem prover PVS. We use an example pro-
tocol used for bus arbitration in asynchronous systems that was studied
by Hooman in [JH94]. The paper presents an approach to protocol verifi-
cation in PVS that greatly reduces the ad-hoc-ness of these verifications.

1 Introduction

In an asynchronous system where many independent devices share a single data
bus, bus contention may occur, where more than one device needs to write to the
data bus at some point in time. In order to resolve the contention, an arbitration
bus is added to the system, which has slightly different characteristics than the
data bus — mostly that simultaneous concurrent writes to the arbitration bus
are allowed1. A protocol is still needed, though, so that the devices engaged in
bus arbitration can determine when the arbitration has terminated and which
device has the right to write to the data bus.

The characteristics of the bus and the protocol are described briefly and
informally in section 2. The descriptions in [JH94] are very short as well, and
we refer the reader to the IEEE 896 FutureBus specification [IEEE88] for full
information. Section 3 shows how we formalize some basic data types that are
used on the bus in PVS. In section 4, we model each individual module as an
automaton in PVS and aggregate this into the global state space of the entire
system. This concludes the model

2 The Arbitration Bus

The arbitration bus has some fixed number K of bus lines. Each of these can
carry one of N different signal levels, giving us NK different possible signal
patterns on the arbitration bus. The devices on the arbitration bus can write
values to their own link lines, which connect the device to the bus itself. The
lines of the arbitration bus individually take on the maximum of the values
written to the corresponding link lines. Figure 1 shows an example bus with
three bus lines, and an example sequence of writes by the devices to their link
lines, along with the effect this has on the value of the arbitration bus.

Every device on the bus has a priority assigned to it. This priority is encoded
in the NK different signal patterns that the arbitration bus can carry. The
arbitration protocol is simple: the device with the highest priority gets the bus.
We will assume that the beginning of the bus contention and arbitration is clear
for all the devices, and that the set of devices contending for the bus is fixed for
the duration of the protocol, i.e. no device joins the arbitration protocol midway
through.

Every device is assumed to have a variable v for computing values needed
in the arbitration protocol. This variable must be able to store the values read
from and written to the arbitration bus. Let P denote the priority value of the
device, and P (j) the value of P in the j-th line. Each device on the arbitration
bus executes the following algorithm:

0. Initially, v = P .

1. Write v to the link lines.
1Typically, in a real hardware device, this is done with tri-state outputs.

l2,2

l2,1

l2,0

Device2

Action Bus Value
- < 000 >
D0 writes < 310 > < 310 >
D2 writes < 003 > < 313 >
D1 writes < 222 > < 323 >
D1 writes < 200 > < 313 >
D2 writes < 000 > < 310 >2

Figure 1: An arbitration bus with K = 3. Bus lines are designated Bk. The
link lines for device i are labeled li,k. The table shows an example sequence of
writes to the bus.

2. Read the arbitration bus and store into v.

3. Compute new value of v as follows:

vj =
{

0 if ∃j′ < j. P (j′) < v(j′)
P (j) otherwise

4. Goto 1.

As shown in [JH94], this protocol reaches a stable state after finite time if
all the modules on the bus complete a full cycle every T time units.

3 Representing Bus Values in PVS

This section introduces some PVS notation and formalizes the notion of “bus
values.”

A description of a system in PVS is divided into a collection of theories.
This is much like the division of a program into modules. A theory may have
parameters, e.g. a theory describing busses may have the bus width as a param-
eter. In the case of the arbitration protocol, we model a bus with K bus lines,
each carrying one of N signal levels, so we give both K and N as parameters to
the theory. Because we will model the bus as a vector of values, we name our
theory NumberVectors as follows:

NumberVectors [K,N : posnat] : THEORY BEGIN

Within a theory we can declare and define functions, and types. We can also
assign types to variables. This assignment merely means that we do not need

to state the type of the variable when it occurs in a function. The following
fragment declares that wherever we use l in the NumberVectors theory it has
type below[K]. The type below[K] is predefined in PVS, and is the range of
natural numbers [0 . . .K − 1].

l : VAR below[K]

We will represent a bus with K lines and N values as a vector of fixed length
K of values in the range [0 . . . N − 1]. A straightforward way to represent such
a vector3 in PVS is as a function with domain below[K] and range below[N];
if v is such a vector, then v(0) denotes the value of the first bus line4. The
following fragment introduces the function type for vectors. The declaration
TYPE+ (as opposed to just TYPE) states that the type is non-empty, and PVS
requires us to prove that there is at least one element of the type.

NumberVectors : TYPE+ = [ below[K] -> below[N] ]

3.1 Ordering Priority Vectors

Since we are interested in the “largest” priority vector in the system, we need
to define an ordering < with respect to which we can determine the largest
vector. The choice of < and the formula in step 3 of the algorithm are critically
interdependent. We choose a lexicographic ordering on the vectors, with bus line
0 being the “most significant” line, and hence the protocol also looks to smaller-
numbered bus lines when computing the value to place on higher-numbered lines.
Our definition of < (there are many equivalent ones) is

<(v1,v2) : bool = EXISTS l : v1(l) < v2(l) AND

FORALL j : j < l IMPLIES v1(j) = v2(j)

Here we see a function definition — the function is <, with two parameters
v1 and v2 which are vectors — returning a truth value (bool in PVS). The
expression defining the value of the function appears after the =. In this case,
the return value is true if and only if v1 comes before v2 lexicographically.

An essential property of this ordering is that it is total, so that there truly
is one “largest” priority vector in any non-empty set of vectors. We can prove
this property by formulating a lemma5 that states this, using the predefined
predicate on relations called strict_total_order? that formalizes this. Our
theorem states that this predicate is true for our <:

ValStrictTotal : LEMMA strict_total_order?(<)

3.2 Computing a new v

The translations into PVS of the formula in step 3 of the algorithm is straight-
forward. We introduce a predicate beaten that states that vector v1 is greater
than vector v2 at some position less than a given j:

3There is a predefined type vector, but it is variable-length.
4We count from zero, and bus line 0 is the first, while bus line K − 1 is the Kth.
5PVS allows many equivalent keywords for theorem: lemma is one of them.

beaten(v1,v2,j) : bool = EXISTS l : l < j AND v1(l) > v2(l)

The rest of step 3 uses computes new values by using this beaten predicate:

comp(v1,v2)(j) : below[N] =

IF beaten(v1,v2,j) THEN 0 ELSE v2(j) ENDIF

Note that the function comp is a curried function with parameters (v1,v2)
and (j). The expression comp(v1,v2) is a function from below[K] to below[N],
i.e. is a NumberVector as already described. This means that a statement like
v = comp(v1,v2) is well-typed if v is a vector.

3.3 Finding the maximum priority

There is of course at least one module in the system whose priority vector is
the largest (two modules might have the same priority vector – this is a system
design error, not a problem with the protocol). We have the function p, which
maps process IDs to priority vectors. We can define the set of module ids with
the largest priority vectors maxids and some element of that set as follows:

maxids : setof[ProcessIds] = { id | FORALL idn : p(id) >= p(idn) }

The set of module ids is of course non-empty, since there must be at least
one largest priority vector in the total order. PVS does not know this, though,
and we will need to prove this fact shortly. The statement

maxid : ProcessIds = choose(maxids)

defines a constant maxid which is any one of the members of that set. This
statement assumes that the set maxids is non-empty, so PVS now generates a
Type Check Condition, TCC, requiring us to prove it.

% Judgement subtype TCC generated (at line 33) for maxids

maxids_nonempty: OBLIGATION nonempty?[ProcessIds](maxids);

We can think of additional properties of the set maxids and state them as
lemmas in PVS. For example, any two module IDs that are in the set maxids
must have the same priority vectors:

maxidsPriority : LEMMA maxids(id) AND maxids(idn) => p(id)=p(idn)

Such a lemma in PVS is considered to be universally quantified over the free
variables in it — hence, this lemma asserts that for all id and idn, if both id
and idn are members of the set maxids, then their priority vectors are equal.

4 Modeling the Protocol as an Automaton in
PVS

In [JH94] little attention is paid to the modeling of the protocol, and it simply
states a number of commitments that the modules make. We will model the

Figure 2: The protocol for a single module, modeled as an automaton.

behavior of each module individually as an automaton that takes the steps
described in section 2. Then we will formalize the state of the entire system
as the product of all the participating modules’ states. Finally, we will use out
PVS automaton framework to create runs of the complete system, so that we
can reason about sequences of actions performed by the system.

Since the protocol as described is a sequence of steps with no control flow
except for the “goto start,” the automaton is straightforward. For every step of
the protocol we introduce a location in the automaton, and a transition out of
the location represents performing the action in that protocol step. This gives us
the automaton shown in figure 2. Note the initial transition which corresponds
to the initialization in step 0 of the protocol.

4.1 Local and Global State

We begin by introducing an enumerated type in PVS so that we can refer to
the locations by name. The PVS statement

Locations : TYPE = {l0, l1, l2}

does this for us. The names introduced are now treated as individual distinct
constants, and we can write, for instance, alias_lo:Locations=l0. Each au-
tomaton (module) individually has a local state consisting of the value of its
link line, the stored value v, and in which location (step of the protocol) it is.
Recall that a module’s priority vector was considered to be part of the global
state of the system. In PVS we will use a record type to collect the state of one
automaton. This gives us

LocalStates : TYPE+ = [# loc : Locations,

v : NumberVectors,

link : NumberVectors #]

In PVS, record accesses can be written as field(recordvalue). Record up-
dates are written as recordvalue WITH [ field := value ].

Now that we have the local state of the modules, we can look at the state
of the system as a whole. We will use a function to map module ids to local
states, just as we used a function to map ids to priority vectors. This yields the
following PVS code:

States : TYPE+ = [ProcessIds -> LocalStates]

4.2 The Arbitration Bus

The value of the arbitration bus itself has not yet been specified at all yet; it
is not part of any module’s local state, nor was it defined in the section about
bus values, since the bus value depends on the local state of the modules (their
link lines). We can calculate the value of the bus given all the local states by
finding the maximum line value per line. This looks like:

maxlinks(s)(j) : (nonempty?[ProcessIds]) =

{ id | FORALL idn : link(s(id))(j) >= link(s(idn))(j) }

maxlink(s)(j) : ProcessIds = choose(maxlinks(s)(j)) ;

bus(s)(j) : below[M] = link(s(maxlink(s)(j)))(j) ;

Note that we first define the set of “greatest link lines”, and then use the fact
that it is nonempty to choose one of those link lines — all of those lines must
have the same value — to assign to the arbitration bus.

4.3 Transitions in the PVS model

Now that the local state of each of the modules has been formalized, and the
global state formalized as the product of all the local states, we need to consider
how the state changes in the system, i.e. what steps each module takes locally
and how it affects the global state of the system.

Clearly each module affects only its own local state — the stored value v,
the location the module is in, as well as the value of the local link lines. The
global effect on the arbitration bus can be computer based on the changed local
state. Each individual module can perform one of three actions, represented
by transitions in the automaton diagram in figure 2, above. Each action is
parameterized by the module number (ProcessId) performing the action, as
follows:

read(id : ProcessId) : read?

compute(id : ProcessId) : compute?

apply(id : ProcessId) : apply?

Obviously, a module can only perform a certain action when it is in the
correct location to do so. We will write a PVS predicate that states this restric-
tion, based on an action a (one of the three just defined) and a global state s,
as follows:

Pre(a)(s) : bool =

CASES a OF

read(id) : loc(s(id)) = l1,

compute(id) : loc(s(id)) = l2,

apply(id) : loc(s(id)) = l0

ENDCASES

Here the CASES statement distinguishes between the three actions just defined;
in each case we check that the location of module id in the given global state
is the right one. s(id) is the local state of module id, and loc() is a function
that returns the module’s location.

Next, when n action is performed, the local state of a module changes. We
can formalize this change in PVS as follows6:

Effect(s0,a,s1) : bool =

CASES a OF

read(id) : s1 = s0 WITH [(id)(v) := bus(s0),

(id)(loc) := l2],

compute(id) : s1 = s0 WITH [(id)(v) := comp(v(s0(id)),p(id)),

(id)(loc):=l0],

apply(id) : s1 = s0 WITH [(id)(link) := v(s0(id)),

(id)(loc):=l1]

ENDCASES

Here the effect of each transition is contained in the WITH expression. Assign-
ments are made to parts of the global state. We use some peculiar PVS syntax7

to assign a new location (loc) and a new value to v or link, depending on the
action.

4.4 Specifying Protocol Runs

A run of the protocol begins with all the modules in their initial location and
then is defined as follows:

• Each action must be permitted in the state it is executed from.

• Each state (except the first) must be the result of the effect of the previous
action on the previous state.

The only missing piece in our formalization is the set of initial states, which
we form as follows:

Init(s) : bool = FORALL id :

loc(s(id))=l0 AND v(s(id))=p(id) AND link(s(id)) = zero

6Since there is no nondeterminism in this system, the formalization is overly general,
but our automaton framework in general needs to use a relation like this one to capture
nondeterministic behavior as well.

7PVS can use WITH notation to update values in records and also in functions; this can also
be Curried, so the given expressions update the function s0 in point id with new field values.

in which we use the special constant zero, which is the all-zero vector of bus
values. Now that we have all of the parts of the formalization of the behavior
of the system, we can use our automaton framework to define the runs of the
system and provide us with some useful PVS machinery to prove properties
about it:

IMPORTING Automata[States, ActionsArb, Init, Pre, Effect]

5 Proving Stability of the Protocol

Since there is no termination condition indicated in the automaton for each
module — the protocol continues regardless — we will not even try to prove
that the protocol terminates, (later, in concluding in section 6, we will discuss
adding a timing constraint so that termination is guaranteed). Instead, we will
show conditions under which all the lines of the arbitration bus reach a stable
value which they keep for the remainder of the execution. When all the lines
of the arbitration bus are stable, the largest priority vector has been found and
we can consider the protocol to have finished.

Our formalization of the notion “a stable bus” is based on counting the lines
of the bus. The bus is stable up to and including line i

We can begin with a collection of trivialities that are needed for technical
reasons. Although the trivialities can be recognized to belong to a class of
similar theorems, formulating the general case yields an extremely unwieldy
theorem which is useless in practice. With additional automation — say, tools
that operate on the automaton representation and not on the PVS encoding
— we could generate these lemmas and prove them mechanically, but for the
purpose of this paper we are stuck with the PVS representation of the automata
and the lemmas we can prove by hand about them.

l0l2Il1 : LEMMA

loc(states(r)(i_1)(id))=l0 AND loc(states(r)(i_1+i_2)(id))=l2 =>

EXISTS (i_3 | i_3<=i_2) : loc(states(r)(i_1+i_3)(id))=l1 ;

l1l0Il2 : LEMMA

loc(states(r)(i_1)(id))=l1 AND loc(states(r)(i_1+i_2)(id))=l0 =>

EXISTS (i_3 | i_3<=i_2) : loc(states(r)(i_1+i_3)(id))=l2 ;

l0l1IApply : LEMMA

loc(states(r)(i_1)(id))=l0 AND loc(states(r)(i_1+i_2)(id))=l1 =>

EXISTS (i_3 | i_3<=i_2) : events(r)(i_1+i_3)=apply(id) ;

5.1 Dividing the Run

We will divide the execution of an automaton into disjunct sub-executions based
on properties of the events and states that occur within those sub executions.
One of the most important properties we use is AtLeastOnce, which states
that in a given run r, a given module with ID id performs a certain action ac
somewhere in the interval (i.e. states and actions numbered) [i0 . . . i1].

AtLeastOnce(r,ac,id)(i_0:nat,i_1:nat | i_0<=i_1) : bool =

EXISTS (i_2:nat | i_0 <= i_2 AND i_2 <= i_1) :

events(r)(i_2)=ac(id)

An immediate application of this property is in defining AllReadApply,
which states that all modules perform a read action and then an apply action
(in that order) in an interval [i0 . . . i1].

AllReadApply(r)(j_1:nat,j_2:nat | j_1<j_2) : bool =

FORALL (id) :

EXISTS (i_1:nat | j_1<=i_1 AND i_1<j_2) :

AtLeastOnce(r,read,id)(j_1,i_1) AND

AtLeastOnce(r,apply,id)(i_1+1,j_2)

Using both of these we define the notion of a “round.” The notion of round is
closely tied to the notion of a stable bus, so we will simply present the definition
here and work toward explaining why it is useful. We define a predicate stating
that a given run r is in round k at step i. The notion of a round is inductively
defined (the run is in round k + 1 at step i if there is a i′ < i such that the run
is in round k at step i′ and some additional property holds between i′ and i),
which means that we need to use the PVS keyword RECURSIVE and provide a
measure to the recursion in order to guarantee that the recursion terminates.
Using the round number k is the obvious thing to do.

Round(r,i,k) : RECURSIVE bool =

IF k=0 THEN AtLeastOnce(r,apply,maxid)(0,i)

ELSE EXISTS (i_2:posnat) : i_2<i AND Round(r,i_2-1,k-1) AND

AllReadApply(r)(i_2,i)

ENDIF MEASURE k ;

5.2 Round 0: The First Line Stabilizes

It is fairly straightforward to conclude that the module with ID maxid has the
largest priority vector and hence the first digit of that vector must be greater
than (or equal to) the first digit of all other priority vectors. From here it can
be concluded that when module maxid writes that first digit to its link line, the
bus takes on that value on line 0.

MaxFirstGreatest : LEMMA p(maxid)(0) >= p(id)(0)

ZS0 : LEMMA

link(states(r)(i)(maxid))(0)=p(maxid)(0) =>

bus(states(r)(i))(0)=p(maxid)(0)

Module maxid will never write anything but p(maxid)(0) to line line 0, and
hence once that value has been written, the bus retains that value on line 0.
Because module maxid must do an apply action to write a value to its link lines,
we can conclude the following (which has an obvious tie-in to the definition of
“round,” above).

ZeroStable : THEOREM

AtLeastOnce(r,apply,maxid)(0,i) =>

BusStable(r,i+1,0) ;

5.3 Stabilizing the Rest of the Bus

Suppose that the bus is stable on lines 0 . . . k (at step i). Then whenever a
module reads from the bus after step i, it will read the values of p(maxid) on
those lines. This means that immediately after module id reads from the bus,
part of the value of v for that module is known. As long as the module stays in
location l2 (i.e. before it computes a new value for v), the value of v remains
unchanged.

AfterReadFromStableBus : LEMMA

BusStable(r,i,k) =>

FORALL (i_1:nat | i<=i_1) : % After i

loc(states(r)(i_1+1)(id))=l2 AND % before new compute

AtLeastOnce(r,read,id)(i,i_1) => % but after reading

(FORALL (k_2: nat | k_2 <= k): % value v is known

v(states(r)(1 + i_1)(id))(k_2) = p(maxid)(k_2)))

Proving this lemma in PVS is non-trivial. We used three additional lemmas
and two abbreviating definitions (which have been unfolded in the formula-
tion above). The proof takes some 90 proof-steps, many of which are PVS-
administrative in nature.

The next step in proving that the bus stabilizes is determining what the
modules compute; in particular what digit k + 1 of the vector v is computed by
each module. We note that the modules use comp() to compute the new values;
since the value of digit k + 1 depends on the values of digits 0 . . . k read from
the bus, it therefore depends on the value of p(maxid) only.

AfterReadComputeFromStableBus : LEMMA

k+1<K AND BusStable(r,i,k) =>

FORALL (j_1:nat | i<=j_1) :

events(r)(j_1)=read(id) =>

FORALL (j_2:nat | j_1<=j_2) :

loc(states(r)(j_2)(id))=l0 =>

v(states(r)(j_2)(id))(k+1) = comp(p(maxid),p(id))(k+1) ;

From this we can conclude that — when the bus is stable up to digit k
— each module will write comp(p(maxid),p(id))(k+1) to its link lines. In
particular, this means that only modules whose priority vector’s digits 0 . . . k
are equal to the digits of p(maxid) will write their digit k + 1 to the link lines.
All other modules will write a 0 to link line k + 1. This means that when every
module has read and written a new value to the link lines, that the digit value
written by maxid is the largest, and the bus takes on this value.

EveryoneComputed : LEMMA

k+1<K =>

( FORALL (id) : link(s(id))(k+1)=comp(p(maxid),p(id))(k+1) ) =>

bus(s)(k+1)=p(maxid)(k+1) ;

Our correctness argument now links the notion of “round” with the stability
of the bus.

Correctness : THEOREM

Round(r,i,k) => BusStable(r,i+1,k) ;

6 Conclusion

In this paper, we have shown how we model the IEEE896 FutureBus bus-
arbitration protocol as a collection of automata in PVS. We subsequently proved
that under the condition that all of the modules (automata) read from and write
to the bus with some regularity (i.e. no module waits indefinitely), the bus sta-
bilizes to the value of the maximum priority. This proof was done without
introducing timing. In order to relate our untimed proof to the original timed
proof in [JH94], we introduce a single timing constraint on the operation of
each module, and derive a time-based termination condition similar8 to that of
[JH94].

Our modeling effort is an example of how protocols can be modeled in PVS
as automata using our automaton framework. Our proof depends, of course, on
the specific protocol and its correctness property, but it also serves as an example
of how proofs based on execution fragments9 can be done. This combination of
automata and execution-fragment based reasoning is nicely supported by our
automaton framework.

Future work:

